Title stata.com

xtrc — Random-coefficients model

Description Quick start Menu

Options Remarks and examples Stored results Methods and formulas

Syntax

References Also see

Description

xtrc fits the Swamy (1970) random-coefficients linear regression model, which does not impose the assumption of constant parameters across panels. Average coefficient estimates are reported by default, but panel-specific coefficients may be requested.

Quick start

Random-coefficients regression of y on x1 and x2 using xtset data

xtrc y x1 x2

Same as above, but report panel-specific best linear predictors

xtrc y x1 x2, betas

Multiple-imputation estimates of random-coefficients regression using mi xtset data

mi estimate: xtrc y x

Menu

Statistics > Longitudinal/panel data > Random-coefficients regression by GLS

```
xtrc depvar indepvars [if] [in] [, options]
```

options	Description
Main	
<u>nocons</u> tant	suppress constant term
<pre>offset(varname)</pre>	include varname in model with coefficient constrained to 1
SE	
vce(vcetype)	<i>vcetype</i> may be conventional, <u>boot</u> strap, or <u>jack</u> knife
Reporting	
$\underline{1}$ evel(#)	set confidence level; default is level(95)
<u>beta</u> s	display group-specific best linear predictors
display_options	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
<u>coefl</u> egend	display legend instead of statistics

A panel variable must be specified; use xtset; see [XT] xtset.

indepvars may contain factor variables; see [U] 11.4.3 Factor variables.

by, collect, mi estimate, and statsby are allowed; see [U] 11.1.10 Prefix commands.

vce(bootstrap) and vce(jackknife) are not allowed with the mi estimate prefix; see [MI] mi estimate. coeflegend does not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Options

Main noconstant, offset(varname); see [R] Estimation options

vce(vcetype) specifies the type of standard error reported, which includes types that are derived from asymptotic theory (conventional) and that use bootstrap or jackknife methods (bootstrap, jackknife); see [XT] vce_options.

vce(conventional), the default, uses the conventionally derived variance estimator for generalized least-squares regression.

Reporting)

level(#); see [R] Estimation options.

betas requests that the group-specific best linear predictors also be displayed.

display_options: noci, nopvalues, noomitted, vsquish, noemptycells, baselevels,
 allbaselevels, nofvlabel, fvwrap(#), fvwrapon(style), cformat(%fmt), pformat(%fmt),
 sformat(%fmt), and nolstretch; see [R] Estimation options.

The following option is available with xtrc but is not shown in the dialog box: coeflegend; see [R] Estimation options.

Remarks and examples

stata.com

In random-coefficients models, we wish to treat the parameter vector as a realization (in each panel) of a stochastic process. xtrc fits the Swamy (1970) random-coefficients model, which is suitable for linear regression of panel data. See Greene (2012, chap. 11) and Poi (2003) for more information about this and other panel-data models.

▶ Example 1

Greene (2012, 1112) reprints data from a classic study of investment demand by Grunfeld and Griliches (1960). In [XT] **xtgls**, we use this dataset to illustrate many of the possible models that may be fit with the xtgls command. Although the models included in the xtgls command offer considerable flexibility, they all assume that there is no parameter variation across firms (the cross-sectional units).

To take a first look at the assumption of parameter constancy, we should reshape our data so that we may fit a simultaneous-equation model with sureg; see [R] sureg. Because there are only five panels here, this is not too difficult.

```
. use https://www.stata-press.com/data/r18/invest2
. reshape wide invest market stock, i(time) j(company)
(j = 1 2 3 4 5)
```

Data	Long	->	Wide
Number of observations Number of variables j variable (5 values) xij variables:	5		20 16 (dropped)
·	invest market stock	->	invest1 invest2 invest5 market1 market2 market5 stock1 stock2 stock5

- . sureg (invest1 market1 stock1) (invest2 market2 stock2)
- > (invest3 market3 stock3) (invest4 market4 stock4) (invest5 market5 stock5)

Seemingly unrelated regression

Equation	Obs	Params	RMSE	"R-squared"	chi2	P>chi2
invest1	20	2	84.94729	0.9207	261.32	0.0000
invest2	20	2	12.36322	0.9119	207.21	0.0000
invest3	20	2	26.46612	0.6876	46.88	0.0000
invest4	20	2	9.742303	0.7264	59.15	0.0000
invest5	20	2	95.85484	0.4220	14.97	0.0006

	Coefficient	Std. err.	z	P> z	[95% conf.	interval]
invest1						
market1	.120493	.0216291	5.57	0.000	.0781007	.1628853
stock1	.3827462	.032768	11.68	0.000	.318522	.4469703
_cons	-162.3641	89.45922	-1.81	0.070	-337.7009	12.97279
invest2						
market2	.0695456	.0168975	4.12	0.000	.0364271	.1026641
stock2	.3085445	.0258635	11.93	0.000	. 2578529	.3592362
_cons	.5043112	11.51283	0.04	0.965	-22.06042	23.06904
invest3						
market3	.0372914	.0122631	3.04	0.002	.0132561	.0613268
stock3	.130783	.0220497	5.93	0.000	.0875663	.1739997
_cons	-22.43892	25.51859	-0.88	0.379	-72.45443	27.57659
invest4						
market4	.0570091	.0113623	5.02	0.000	.0347395	.0792788
stock4	.0415065	.0412016	1.01	0.314	0392472	.1222602
_cons	1.088878	6.258805	0.17	0.862	-11.17815	13.35591
invest5						
market5	.1014782	.0547837	1.85	0.064	0058958	.2088523
stock5	.3999914	.1277946	3.13	0.002	.1495186	.6504642
_cons	85.42324	111.8774	0.76	0.445	-133.8525	304.6989

Here we instead fit a random-coefficients model:

- . use https://www.stata-press.com/data/r18/invest2
- . xtrc invest market stock

Random-coefficients regression Number of obs 100 Number of groups = Group variable: company

Time variable: time Obs per group:

> min = 20 20.0 avg = max =20 Wald chi2(2) = 17.55 Prob > chi2 = 0.0002

invest	Coefficient	Std. err.	z	P> z	[95% conf.	interval]
market	.0807646	.0250829	3.22	0.001	.0316031	.1299261
stock	.2839885	.0677899	4.19	0.000	.1511229	.4168542
_cons	-23.58361	34.55547	-0.68	0.495	-91.31108	44.14386

Test of parameter constancy: chi2(12) = 603.99 Prob > chi2 = 0.0000 Just as the results of our simultaneous-equation model do not support the assumption of parameter constancy, the test included with the random-coefficients model also indicates that the assumption is not valid for these data. With large panel datasets, we would not want to take the time to look at a simultaneous-equations model (aside from the fact that our doing so was subjective).

4

Stored results

xtrc stores the following in e():

```
Scalars
                              number of observations
    e(N)
    e(N_g)
                              number of groups
    e(df_m)
                              model degrees of freedom
    e(chi2)
                              \chi^2 for comparison test
    e(chi2_c)
    e(df_chi2c)
                              degrees of freedom for comparison \chi^2 test
                              smallest group size
    e(g_min)
    e(g_avg)
                              average group size
    e(g_{max})
                              largest group size
    e(rank)
                              rank of e(V)
Macros
    e(cmd)
                              xtrc
    e(cmdline)
                              command as typed
    e(depvar)
                              name of dependent variable
                              variable denoting groups
    e(ivar)
    e(tvar)
                              variable denoting time within groups
    e(title)
                              title in estimation output
    e(offset)
                              linear offset variable
    e(chi2type)
                              Wald; type of model \chi^2 test
                              vcetype specified in vce()
    e(vce)
    e(properties)
                              program used to implement predict
    e(predict)
    e(marginsnotok)
                              predictions disallowed by margins
    e(asbalanced)
                              factor variables fyset as asbalanced
                              factor variables fyset as asobserved
    e(asobserved)
Matrices
                              coefficient vector
    e(b)
                              \Sigma matrix
    e(Sigma)
    e(beta_ps)
                              matrix of best linear predictors
                              variance-covariance matrix of the estimators
    e(V)
    e(V_ps)
                              matrix of variances for the best linear predictors; row i contains vec of
                                variance matrix for group i predictor
Functions
    e(sample)
                              marks estimation sample
```

In addition to the above, the following is stored in r():

```
Matrices
r(table)
```

matrix containing the coefficients with their standard errors, test statistics, p-values, and confidence intervals

Note that results stored in r() are updated when the command is replayed and will be replaced when any r-class command is run after the estimation command.

6

Methods and formulas

In a random-coefficients model, the parameter heterogeneity is treated as stochastic variation. Assume that we write

$$\mathbf{y}_i = \mathbf{X}_i \boldsymbol{\beta}_i + \boldsymbol{\epsilon}_i$$

where i = 1, ..., m, and β_i is the coefficient vector $(k \times 1)$ for the ith cross-sectional unit, such that

$$\boldsymbol{\beta}_i = \boldsymbol{\beta} + \boldsymbol{\nu}_i \qquad E(\boldsymbol{\nu}_i) = \mathbf{0} \qquad E(\boldsymbol{\nu}_i \boldsymbol{\nu}_i') = \boldsymbol{\Sigma}$$

Our goal is to find $\widehat{\beta}$ and $\widehat{\Sigma}$.

The derivation of the estimator assumes that the cross-sectional specific coefficient vector β_i is the outcome of a random process with mean vector β and covariance matrix Σ ,

$$\mathbf{y}_i = \mathbf{X}_i \boldsymbol{\beta}_i + \boldsymbol{\epsilon}_i = \mathbf{X}_i (\boldsymbol{\beta} + \boldsymbol{\nu}_i) + \boldsymbol{\epsilon}_i = \mathbf{X}_i \boldsymbol{\beta} + (\mathbf{X}_i \boldsymbol{\nu}_i + \boldsymbol{\epsilon}_i) = \mathbf{X}_i \boldsymbol{\beta} + \boldsymbol{\omega}_i$$

where $E(\boldsymbol{\omega}_i) = \mathbf{0}$ and

$$E(\boldsymbol{\omega}_{i}\boldsymbol{\omega}_{i}') = E\left\{ (\mathbf{X}_{i}\boldsymbol{\nu}_{i} + \boldsymbol{\epsilon}_{i})(\mathbf{X}_{i}\boldsymbol{\nu}_{i} + \boldsymbol{\epsilon}_{i})' \right\} = E(\boldsymbol{\epsilon}_{i}\boldsymbol{\epsilon}_{i}') + \mathbf{X}_{i}E(\boldsymbol{\nu}_{i}\boldsymbol{\nu}_{i}')\mathbf{X}_{i}' = \sigma_{i}^{2}\mathbf{I} + \mathbf{X}_{i}\boldsymbol{\Sigma}\mathbf{X}_{i}' = \boldsymbol{\Pi}_{i}$$

Stacking the m equations, we have

$$y = X\beta + \omega$$

where $\Pi \equiv E(\omega \omega')$ is a block diagonal matrix with Π_i , i = 1...m, along the main diagonal and zeros elsewhere. The GLS estimator of $\hat{\beta}$ is then

$$\widehat{\boldsymbol{\beta}} = \left(\sum_i \mathbf{X}_i' \mathbf{\Pi}_i^{-1} \mathbf{X}_i\right)^{-1} \sum_i \mathbf{X}_i' \mathbf{\Pi}_i^{-1} \mathbf{y}_i = \sum_{i=1}^m \mathbf{W}_i \mathbf{b}_i$$

where

$$\mathbf{W}_i = \left\{ \sum_{i=1}^m (\mathbf{\Sigma} + \mathbf{V}_i)^{-1} \right\}^{-1} (\mathbf{\Sigma} + \mathbf{V}_i)^{-1}$$

 $\mathbf{b}_i = (\mathbf{X}_i' \mathbf{X}_i)^{-1} \mathbf{X}_i' \mathbf{y}_i$ and $\mathbf{V}_i = \sigma_i^2 (\mathbf{X}_i' \mathbf{X}_i)^{-1}$, showing that the resulting GLS estimator is a matrix-weighted average of the panel-specific OLS estimators. The variance of $\widehat{\boldsymbol{\beta}}$ is

$$\operatorname{Var}(\widehat{\boldsymbol{\beta}}) = \sum_{i=1}^{m} (\boldsymbol{\Sigma} + \mathbf{V}_i)^{-1}$$

To calculate the above estimator $\hat{\beta}$ for the unknown Σ and V_i parameters, we use the two-step approach suggested by Swamy (1970):

$$\begin{split} \mathbf{b}_i &= \text{OLS panel-specific estimator} \\ \widehat{\sigma}_i^2 &= \frac{\widehat{\epsilon}_i' \widehat{\epsilon}_i}{n_i - k} \\ \widehat{\mathbf{V}}_i &= \widehat{\sigma}_i^2 \left(\mathbf{X}_i' \mathbf{X}_i \right)^{-1} \\ \overline{\mathbf{b}} &= \frac{1}{m} \sum_{i=1}^m \mathbf{b}_i \\ \widehat{\boldsymbol{\Sigma}} &= \frac{1}{m-1} \left(\sum_{i=1}^m \mathbf{b}_i \mathbf{b}_i' - m \overline{\mathbf{b}} \, \overline{\mathbf{b}}' \right) - \frac{1}{m} \sum_{i=1}^m \widehat{\mathbf{V}}_i \end{split}$$

The two-step procedure begins with the usual OLS estimates of β_i . With those estimates, we may proceed by obtaining estimates of $\widehat{\mathbf{V}}_i$ and $\widehat{\mathbf{\Sigma}}$ (and thus $\widehat{\mathbf{W}}_i$) and then obtain an estimate of β .

Swamy (1970) further points out that the matrix $\widehat{\Sigma}$ may not be positive definite and that because the second term is of order 1/(mT), it is negligible in large samples. A simple and asymptotically expedient solution is simply to drop this second term and instead use

$$\widehat{\Sigma} = \frac{1}{m-1} \left(\sum_{i=1}^{m} \mathbf{b}_{i} \mathbf{b}'_{i} - m\overline{\mathbf{b}} \overline{\mathbf{b}}' \right)$$

As discussed by Judge et al. (1985, 541), the feasible best linear predictor of β_i is given by

$$\begin{split} \widehat{\boldsymbol{\beta}}_i &= \widehat{\boldsymbol{\beta}} + \widehat{\boldsymbol{\Sigma}} \mathbf{X}_i' \left(\mathbf{X}_i \widehat{\boldsymbol{\Sigma}} \mathbf{X}_i' + \widehat{\boldsymbol{\sigma}}_i^2 \mathbf{I} \right)^{-1} \left(\mathbf{y}_i - \mathbf{X}_i \widehat{\boldsymbol{\beta}} \right) \\ &= \left(\widehat{\boldsymbol{\Sigma}}^{-1} + \widehat{\mathbf{V}}_i^{-1} \right)^{-1} \left(\widehat{\boldsymbol{\Sigma}}^{-1} \widehat{\boldsymbol{\beta}} + \widehat{\mathbf{V}}_i^{-1} \mathbf{b}_i \right) \end{split}$$

The conventional variance of $\widehat{\beta}_i$ is given by

$$\mathrm{Var}(\widehat{\boldsymbol{\beta}}_i) = \mathrm{Var}(\widehat{\boldsymbol{\beta}}) + (\mathbf{I} - \mathbf{A}_i) \left\{ \widehat{\mathbf{V}}_i - \mathrm{Var}(\widehat{\boldsymbol{\beta}}) \right\} (\mathbf{I} - \mathbf{A}_i)'$$

where

$$\mathbf{A}_i = \left(\widehat{\mathbf{\Sigma}}^{-1} + \widehat{\mathbf{V}}_i^{-1}\right)^{-1} \widehat{\mathbf{\Sigma}}^{-1}$$

To test the model, we may look at the difference between the OLS estimate of β , ignoring the panel structure of the data and the matrix-weighted average of the panel-specific OLS estimators. The test statistic suggested by Swamy (1970) is given by

$$\chi^2_{k(m-1)} = \sum_{i=1}^m (\mathbf{b}_i - \overline{\boldsymbol{\beta}}^*)' \widehat{\mathbf{V}}_i^{-1} (\mathbf{b}_i - \overline{\boldsymbol{\beta}}^*) \quad \text{where} \quad \overline{\boldsymbol{\beta}}^* = \left(\sum_{i=1}^m \widehat{\mathbf{V}}_i^{-1}\right)^{-1} \sum_{i=1}^m \widehat{\mathbf{V}}_i^{-1} \mathbf{b}_i$$

Johnston and DiNardo (1997) have shown that the test is algebraically equivalent to testing

$$H_0: \boldsymbol{\beta}_1 = \boldsymbol{\beta}_2 = \dots = \boldsymbol{\beta}_m$$

in the generalized (groupwise heteroskedastic) xtgls model, where V is block diagonal with ith diagonal element Π_i .

References

Eberhardt, M. 2012. Estimating panel time-series models with heterogeneous slopes. Stata Journal 12: 61-71.

Greene, W. H. 2012. Econometric Analysis. 7th ed. Upper Saddle River, NJ: Prentice Hall.

Grunfeld, Y., and Z. Griliches. 1960. Is aggregation necessarily bad? Review of Economics and Statistics 42: 1–13. https://doi.org/10.2307/1926089.

Johnston, J., and J. DiNardo. 1997. Econometric Methods. 4th ed. New York: McGraw-Hill.

Judge, G. G., W. E. Griffiths, R. C. Hill, H. Lütkepohl, and T.-C. Lee. 1985. The Theory and Practice of Econometrics. 2nd ed. New York: Wiley.

Nichols, A. 2007. Causal inference with observational data. Stata Journal 7: 507-541.

Poi, B. P. 2003. From the help desk: Swamy's random-coefficients model. Stata Journal 3: 302-308.

Swamy, P. A. V. B. 1970. Efficient inference in a random coefficient regression model. *Econometrica* 38: 311–323. https://doi.org/10.2307/1913012.

——. 1971. Statistical Inference in Random Coefficient Regression Models. New York: Springer.

Also see

- [XT] **xtrc postestimation** Postestimation tools for xtrc
- [XT] **xtreg** Fixed-, between-, and random-effects and population-averaged linear models⁺
- [XT] **xtset** Declare data to be panel data
- [ME] **mixed** Multilevel mixed-effects linear regression
- [MI] Estimation Estimation commands for use with mi estimate
- [U] 20 Estimation and postestimation commands

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.